
Appendix: Generating New Cases with SMOTE
 
ods _all_ close;
proc modeclus data=Input (where=(target=1)) dk=11 /*get 10 nearest neighbors */
neighbor out=modeclus_out;
var var_: ; /*Specify the input numeric variables here*/
id obs_id ; /*Sepcify the unique id variable of each observation*/
ods output Neighbor=nTest; /*output the nearest neighbors' id along with their distance*/
run;
ods listing;

/* List each pair of a rare case and its 10 nearest neighbors*/
data neighbor_id (keep=center_id nbor index);
set ntest(where=(distance<3));/*Optional: set a distance threshold*/
retain center_id '000000000000';
index=_N_; /*prepare id for the new cases*/
if not missing(id) then center_ID = put(id,$12.);
else center_id=center_id;
run;


/* transpose the raw data*/
proc transpose data=Input(where=(target=1)) out=work.raw1 prefix=ID_;
var var_:;
id obs_id ;
run;

/*Generating random cases with look up table*/
data _NULL_;
set NEIGHBOR_ID end=eof;
if not eof then do;
	call execute 
		( 'DATA raw1 ; ');
	call execute 
		( 'set raw1  ;' ) ;
	call execute
		("_R_=rand('uniform');"); 
	call execute
		( 'new_'||trim(left((index)))|| ' = (id_'||trim(left((nbor)))||' - id_'||trim(left((center_id)))||') * _R_ ');
call execute( ' + ID_'||trim(left((center_id)))||' ;');
call execute
		('run;');
	end;
if eof then do;
call execute
	('data work.raw2 (keep=_NAME_ new:);  set raw1; run; ');
end;
run;

/* transpose the new cases back to the original layout*/
proc transpose data = work.raw2 out=new_cases;
id _NAME_;
var _NUMERIC_;
run;

